Research Note 85-77

AD-A160 560

SHOT GROUP ANALYSIS PROGRAM FOR THE M16A1 RIFLE

Beth W. Thomas and James E. Schroeder

ARI Field Unit at Fort Benning, Georgia
Seward Smith, Chief

TRAINING RESEARCH LABORATORY
Harold F. O'Neil, Jr., Director

U. S. Army

Research Institute for the Behavioral and Social Sciences

July 1985

U. S. ARMY RESEARCH INSTITUTE
FOR THE BEHAVIORAL AND SOCIAL SCIENCES

A Field Operating Agency under the Jurisdiction of the

Deputy Chief of Staff for Personnel

EDGAR M. JOHNSON
Technical Director

WM. DARRYL HENDERSON
COL, IN
Commanding

Unclassified

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

REPORT DOCUMENTATION PAGE

READ INSTRUCTIONS BEFORE COMPLETING FORM

1. REPORT NUMBER: ARI Research Note 85-77	2. GOVT ACCESSION NO.: AD-A160 530
3. RECIPIENT'S CATALOG NUMBER:	
4. TITLE (and Subtitle): Shot Group Analysis Program for the M16A1 Rifle	5. TYPE OF REPORT & PERIOD COVERED: July 84 - Jan 85
	6. PERFORMING ORG. REPORT NUMBER:
7. AUTHOR(s): Beth W. Thomas, James E. Schroeder	8. CONTRACT OR GRANT NUMBER(s): NA
9. PERFORMING ORGANIZATION NAME AND ADDRESS: ARI Fort Benning Field Unit, P.O. Box 2086, Fort Benning, GA 31905-0686	10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS: 2Q263743A794, 5710 322
11. CONTROLLING OFFICE NAME AND ADDRESS: US Army Research Institute for the Behavioral and Social Sciences, 5001 Eisenhower Avenue, Alexandria, VA 22333-5600	12. REPORT DATE: July 1985
	13. NUMBER OF PAGES: 16
14. MONITORING AGENCY NAME & ADDRESS (if different from Controlling Office)	15. SECURITY CLASS. (of this report): Unclassified
	15a. DECLASSIFICATION/DOWNGRADING SCHEDULE

16. DISTRIBUTION STATEMENT (of this Report)

Approved for public release; distribution unlimited.

17. DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report)

18. SUPPLEMENTARY NOTES

19. KEY WORDS (Continue on reverse side if necessary and identify by block number)

marksmanship
M16A1 rifle
ballistics
shot group

20. ABSTRACT (Continue on reverse side if necessary and identify by block number)

This note contains a description of and BASIC program listing for a shot group analysis program for microcomputers. Using this program, target coordinates of individual shots are entered and important shot group statistics are calculated (e.g., mean and median radial error from target center and shot group center, standard deviations of X and Y, etc.). These measures can serve as valuable dependent variables in the research and evaluation of weapons, ammunition, and marksmanship training.

DD FORM 1 JAN 73 1473 EDITION OF 1 NOV 65 IS OBSOLETE

Unclassified

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

SHOT GROUP ANALYSIS PROGRAM FOR THE M16A1 RIFLE

INTRODUCTION

In marksmanship training and evaluation, scientists and trainers are often forced to use crude perfomance measures such as hit or miss. New technological advances offer the great promise of precise individual live-fire shot location along with any other desired statistical manipulations performed quickly by a computer. This new technology has been named Location of Miss and Hit (LOMAH) and offers tomorrow's scientists and trainers tremendous information gathering, analysis, and storage capabilities. However, LOMAH equipment is not currently available in sufficient quantities to help today's Army's scientists and trainers. Hence, there is a need for a way to quickly assess the quality of marksmanship performance based on input from more traditional methods (e.g., from the distribution and placement of bullet holes in paper targets). The SHOTGROUP program listed in Appendix A provides that capability.

HOW TO USE THIS PROGRAM

The SHOTGROUP program was written in BASIC on an Apple IIe microcomputer. However, it should be relatively easy to adapt to other microcomputers with BASIC capabilities. The user must provide the X and Y coordinates of the "center of mass" or other aiming point, along with the X and Y coordinates for each individual shot. These coordinates would presumably be taken from actual targets, scorecard representations, simulator scores, or from LOMAH equipment. The coordinate system used can be arbitrary, depending on how the results are to be used. However, it is suggested that the units used on the grid matrix be measured M16A1 sight "clicks." A M16A1 rifle sight click is equal to one minute of angle. For example, if the soldier is firing at 25m scaled targets, then each unit on the coordinate grid would be .28 in. because one minute of angle at 25m is .28 in. If a given shot on a 25m scaled target were 3.5 in. high and 2.8 in. left, then the coordinates for that shot would be X = -10 and Y = +12.5 (using the X = 0, Y = 0 origin as the center of the target). Similarly, for a 50 m target each coordinate unit would be .56 in.; for a 75m target each coordinate unit would be .84 in.; etc. A plastic template which is appropriate for 25m targets has been included as Figure 1. This template can be quickly placed over a 25m scaled target. Then, individual shot coordinates can be read, recorded, and/or used as data entries for the computer program listed in Appendix A. Also shown in Figure 1 are three example shot groups. Analyses of these three shot groups are shown in Tables 1-3.



Figure 1. Grid template for converting gun scaled sighted shot location coordinates. Each grid unit equals one Mil (1 sight click or one minute of angle).


Example Group #1 (see Table 1)
Example Group #2 (see Table 2)
Center of Target
Example Group #3 (see Table 3)

Table 1

Example SHOTGROUP Print-Out #1: Three-Round Group

SUBJECT CODE: 001
NUMBER OF ROUNDS: 3
TARGET CENTER OF MASS: (0,0)

SHOT DATA:

SHOT	X	Y	RADIAL ERROR (TARGET CENTER)	RADIAL DISTANCE (SHOT GROUP CENTER)
1	4	7	8.06	2.83
2	9	2	9.22	4.24
3	5	6	7.81	1.41

MEAN X = 6 MEAN Y = 5
STANDARD DEVIATION X = 2.65 STANDARD DEVIATION Y = 2.65
SUM OF X AND Y VARIANCES: 14.05

SHOT GROUP DATA:

RADIAL ERROR (TARGET CENTER): MEAN = 8.36 MEDIAN = 8.06

RADIAL DISTANCE (SHOT GROUP CENTER): MEAN = 2.83 MEDIAN = 2.83

COORDINATES OF CENTER OF SHOT GROUP : (6 , 5)
EXTREME SPREAD (SHOT 1 AND SHOT 2) = 7.07
DISTANCE OF CENTER OF SHOT GROUP FROM CENTER OF MASS: 7.81
ANGLE OF SHOT GROUP FROM CENTER OF MASS IN DEGREES: 39.81
SHOT GROUP CENTER IS LOCATED IN QUADRANT: 1

****

Table 2

Example SHOTGROUP Print-Out #2: Four-Round Group

SUBJECT CODE: 002
NUMBER OF ROUNDS: 4
TARGET CENTER OF MASS: (0,0)

SHOT DATA:

SHOT	X	Y	RADIAL ERROR (TARGET CENTER)	RADIAL DISTANCE (SHOT GROUP CENTER)
1	-7	4	8.06	.56
2	-9	2	9.22	2.7
3	-6	3	6.71	.9
4	-5	5	7.07	2.3

MEAN X = -6.75 MEAN Y = 3.5
STANDARD DEVIATION X = 1.71 STANDARD DEVIATION Y = 1.29
SUM OF X AND Y VARIANCES: 4.59

SHOT GROUP DATA:

RADIAL ERROR (TARGET CENTER): MEAN = 7.77 MEDIAN = 7.57

RADIAL DISTANCE (SHOT GROUP CENTER): MEAN = 1.62 MEDIAN = 1.6

COORDINATES OF CENTER OF SHOT GROUP : (-6.75 , 3.5)
EXTREME SPREAD (SHOT 2 AND SHOT 4) = 5
DISTANCE OF CENTER OF SHOT GROUP FROM CENTER OF MASS: 7.6
ANGLE OF SHOT GROUP FROM CENTER OF MASS IN DEGREES: 152.59
SHOT GROUP CENTER IS LOCATED IN QUADRANT: 2

****

Table 3

Example SHOTGROUP Print-Out #3: Six-Round Group

SUBJECT CODE: 003
NUMBER OF ROUNDS: 6
TARGET CENTER OF MASS: (0,0)

SHOT DATA:

SHOT	X	Y	RADIAL ERROR (TARGET CENTER)	RADIAL DISTANCE (SHOT GROUP CENTER)
1	2	3	3.61	6.77
2	-3	1	3.16	3.43
3	2	-3	3.61	4.22
4	-4	-5	6.4	3.24
5	-6	-3	6.71	3.89
6	-4	-7	8.06	5.02

MEAN X = -2.17 MEAN Y = -2.33
STANDARD DEVIATION X = 3.37 STANDARD DEVIATION Y = 3.72
SUM OF X AND Y VARIANCES: 25.2

SHOT GROUP DATA:

RADIAL ERROR (TARGET CENTER): MEAN = 5.26 MEDIAN = 5

RADIAL DISTANCE (SHOT GROUP CENTER): MEAN = 4.43 MEDIAN = 4.06

COORDINATES OF CENTER OF SHOT GROUP : (-2.17 , -2.33)
EXTREME SPREAD (SHOT 1 AND SHOT 6) = 11.66
DISTANCE OF CENTER OF SHOT GROUP FROM CENTER OF MASS: 3.18
ANGLE OF SHOT GROUP FROM CENTER OF MASS IN DEGREES: 227.04
SHOT GROUP CENTER IS LOCATED IN QUADRANT: 3

****

The purpose of SHOTGROUP is to provide statistical summary data for a given shot group (e.g., measures of accuracy and dispersion). Hence, the coordinates of at least two shots must be captured and entered into the program. When the program is loaded and run, the user will first be asked to provide some general information:

DO YOU WANT A PRINT-OUT? (Results are sent to the printer or monitor.)

CENTER OF MASS X? (What is the X value at the center of the target?)

CENTER OF MASS Y? (What is the Y value at the center of the target?)

HOW MANY SOLDIERS? (How many shot groups will you be entering?)

The program will next ask for information about a given shot group:

SUBJECT CODE? (Enter identifier for this shot group.)

NUMBER OF ROUNDS? (How many shots in this group?)

SHOT 1 X= (Enter X coordinate for first shot.)

SHOT 1 Y= (Enter Y coordinate for first shot.)

SHOT 2 X= (Enter X coordinate for second shot.)

SHOT 2 Y= (Enter Y coordinate for second shot.)

ETC... (NOTE: Assignment of shot "number" is usually arbitrary.)

After all the shot coordinates are entered, the program will list them and ask the user to indicate whether they are correct. If the user indicates the data are correct, then the program will complete the analysis and print the results. If the user indicates an error, the program will ask which shot data is incorrect and then ask for new coordinates. This correction routine will continue until the user indicates that all the data are correct.

PROGRAM OUTPUT

Tables 1-3 show three example shot-group analyses. These three shot groups are shown in Figure 1. The summary data for each shot group consists of:

1. SUBJECT CODE, NUMBER OF ROUNDS, AND COORDINATES OF CENTER OF MASS (The "center of mass" is the same as "center of target.")

2. SHOT COORDINATES, RADIAL ERROR, AND RADIAL DISTANCE for each shot. (NOTE: RADIAL ERROR is defined as the distance from a given shot to the center of the target and should be considered a measure of "accuracy" while RADIAL DISTANCE is defined as the distance of a given shot from the center of the shot group and should be considered a measure of consistency or dispersion.)

3. MEAN X AND MEAN Y FOR THE SHOT GROUP.

4. STANDARD DEVIATIONS OF X AND Y AND THE SUM OF THE VARIANCES OF X AND Y FOR THE SHOT GROUP. (NOTE: These statistics are important marksmanship indices of the consistency or dispersion of the shot group.)

5. MEAN AND MEDIAN RADIAL ERROR (NOTE: These statistics are measures of accuracy, i.e., how close is the shot group to the center of the target?)

6. MEAN AND MEDIAN RADIAL DISTANCE. (NOTE: These measures are important marksmanship indices of consistency or "tightness" of the shot group.)

7. COORDINATES OF THE CENTER OF THE SHOT GROUP (Note: Same as the MEAN X and MEAN Y of the shot group.)

8. EXTREME SPREAD (NOTE: EXTREME SPREAD is often used in marksmanship as a quick index of consistency or dispersion. It is defined as the largest distance between any two shots. In this program, both individual shots are identified and the distance is reported.)

9. DISTANCE OF CENTER OF SHOT GROUP FROM CENTER OF MASS (NOTE: This is an index of "accuracy", i.e., how far the shot group is from the center of the target.

10. ANGLE OF SHOT GROUP AND QUADRANT. (NOTE: The angle is reported in degrees (0-359) and the QUADRANT follows normal conventions: QUADRANT 1 = 1 to 89 degrees; QUADRANT 2 = 91 to 179 degrees; QUADRANT 3 = 181 to 269 degrees; and QUADRANT 4 = 271 to 359 degrees.)

POSSIBLE USES FOR THIS PROGRAM

Given the inherent variability in any person-weapon system, location of a single shot relays relatively little information about performance. However, charactersistics of the entire shot group can contain a great deal of information. An excellent marksman must be able to: a) shoot a consistent or "tight" shot group and b) place the center of that shot group on the center of the target or point-of-aim. Of these two, the former is probably a more important measure because in order to obtain a tight shot group, the man/weapon system must show consistency in steady position, trigger manipulation, and point-of-aim. If a person/weapon system is unable to shoot a tight shot group, then there is something wrong with the person/weapon system.

Given proper weapon quality control, the problem can usually be attributed to the person. If a person is able to shoot a tight shot group, but that shot group doesn't fall on the center of the target, then simple sight adjustments can be made to "zero" the weapon. While knowledge about how and when to adjust sights is an important cognitive/mechanical component to marksmanship, it cannot be considered to be the more important measure of marksmanship skill. For this reason, measures of dispersion of the shot group such as STANDARD DEVIATION OF X AND Y, SUM OF THE VARIANCES OF X D Y, and MEAN OR MEDIAN RADIAL DISTANCE, are often used as dependent measures of marksmanship skill, especially early in training before the person has had an opportunity to adjust sights and zero the weapon.

The use of this program allows scientists and trainers to obtain measures of both "accuracy" (MEAN/MEDIAN RADIAL ERROR) and dispersion (STANDARD DEVIATION OF X or Y, SUM OF X AND Y VARIANCES, and MEAN/MEDIAN RADIAL DISTANCE). These measures can be used by the scientist as dependent variables for correlational or experimental analyses. These measures can be used separately or jointly by the trainer as an index of marksmanship ability. For example, it could be argued that the person/M16A1 system should be capable of placing 70% of the rounds in a 4 cm circle at 25m (4 cm at 25 m is trigonometrically equivalent to the width of an "E" type personnel target at 300 m). Using this criterion, a cut-off value can be calculated and the following rule can be formed: If the STANDARD DEVIATION OF X or Y is greater than 2.7 clicks, then the person should probably recieve remediation and/or the weapon should be checked. Also, if the MEAN/MEDIAN RADIAL DISTANCE is much smaller than the MEAN/MEDIAN RADIAL ERROR, then the person needs to adjust sights and zero the weapon.

APPENDIX A

SHOTGROUP Program Listing

SHOTGROUP PROGRAM

LIST1-450

```
10  REM *** SHOT GROUP ANALYSIS PROGRAM ***
20  REM
30  REM *** ARI RESEARCH PRODUCT ***
40  REM *** BY BETH THOMAS AND JAMES E. SCHROEDER ***
50  REM
60  REM *** THIS VERSION WILL RUN ON AN APPLE II+ OR IIE ***
70  REM  *** THE PROGRAM CONDUCTS A SHOT GROUP ANALYSIS FROM INPUT SHOT COORDINATES ***
80  REM  *** NOTE:  IN THIS PROGRAM, RADIAL ERROR IS DISTANCE FROM A SHOT TO THE ***
90  REM  *** CENTER OF THE TARGET AND RADIAL DISTANCE REFERS TO THE DISTANCE FROM A ***
100  REM  *** SHOT TO THE CENTER OF THE SHOT GROUP ***
110  REM *********************************
120  REM
130  PRINT "DO YOU WANT A PRINT-OUT, Y OR N?";P$
140  GET P$
150  DIM X(100),Y(100),RE(100),ROE(100),RD(100),TR(100)
160  REM ************************************************************************
170  REM *** PRELIMINARY INPUT OF INFORMATION (REMAINS CONSTANT FOR ENTIRE RUN ***
180  REM ************************************************************************
190  PRINT "CENTER OF MASS X?"
200  INPUT XCM
210  PRINT "CENTER OF MASS Y?"
220  INPUT YCM
230  PRINT "HOW MANY SOLDIERS?"
240  INPUT J
250  REM ***************************************
260  REM *** MAJOR LOOP FOR EACH SHOT GROUP ***
270  REM ***************************************
280  FOR D = 1 TO J
290  PRINT "SUBJECT CODE?"
300  INPUT SC$
310  PRINT "NUMBER OF ROUNDS?"
320  INPUT N
330  IF N >  = 2 GOTO 390
340  PRINT "THIS PROGRAM WILL NOT WORK FOR N=1"
350  GOTO 310
360  REM ********************************************
370  REM *** SHOT COORDINATES INPUT AND CORRECTION ***
380  REM ********************************************
390  FOR I = 1 TO N
400  PRINT "SHOT";(I);"   X = ": INPUT X(I)
410  PRINT "SHOT";(I);"   Y = ": INPUT Y(I)
420  NEXT I
430  FOR I = 1 TO N
440  PRINT "SHOT";(I);"   X =";X(I);"   Y =";Y(I)
450  NEXT I
```

```
LIST460-880

460  PRINT "ARE THESE CORRECT?"
470  GET PC$
480  IF PC$ <  > "N" GOTO 560
490  PRINT "CHANGE WHICH SHOT?": INPUT K
500  PRINT "SHOT";(K);"   X =  ": INPUT X(K)
510  PRINT "SHOT";(K);"   Y =  ": INPUT Y(K)
520  GOTO 430
530  REM ****************************************
540  REM *** BASIC STATISTICS FOR RADIAL ERROR ***
550  REM ****************************************
560 XSUM = 0
570 YSUM = 0
580 SUMRE = 0
590  FOR I = 1 TO N
600 RE(I) = (((X(I) - XCM) © 2) + ((Y(I) - YCM) © 2)) © .5
610 ROE(I) = RE(I)
620 ROE(I) = ( INT ((ROE(I) + .005) * 100)) / 100
630 XSUM = X(I) + XSUM
640 YSUM = Y(I) + YSUM
650 SUMRE = RE(I) + SUMRE
660  NEXT I
670 MNRE = SUMRE / N
680 MNRE = ( INT ((MNRE + .005) * 100)) / 100
690 MX = XSUM / N
700 MX = ( INT ((MX + .005) * 100)) / 100
710 MY = YSUM / N
720 MY = ( INT ((MY + .005) * 100)) / 100
730  REM ************************************
740  REM *** DETERMINE MEDIAN RADIAL ERROR ***
750  REM ************************************
760  FOR I = N TO 2 STEP  - 1
770  FOR K = 1 TO N - 1
780  IF RE(K) < RE(K + 1) GOTO 820
790 A = RE(K)
800 RE(K) = RE(K + 1)
810 RE(K + 1) = A
820  NEXT K
830  NEXT I
840  IF N > 2 * ( INT (N / 2)) GOTO 870
850 MEDRE = ((RE(N / 2)) + (RE((N / 2) + 1))) / 2
860  GOTO 880
870 MEDRE = RE((N / 2) + .5)
880 MEDRE = ( INT ((MEDRE + .005) * 100)) / 100
```

```
LIST890-1350

890  REM ******************************
900  REM *** COMPUTE EXTREME SPREAD ***
910  REM ******************************
920 LO = 0:HI = 0:EXSP = 0
930  FOR I = 1 TO N - 1
940 L = I + 1
950  FOR J = L TO N
960 DI = (((X(I) - X(J)) © 2) + ((Y(I) - Y(J)) © 2)) © .5
970  IF DI > EXSP GOTO 990
980  GOTO 1020
990 LO = I
1000 HI = J
1010 EXSP = DI
1020  NEXT J
1030  NEXT I
1040 EXSP = ( INT ((EXSP + .005) * 100)) / 100
1050  REM ************************************************************************
1060  REM *** BASIC STATISTICS FOR RADIAL DISTANCE FROM CENTER OF SHOT GROUP ***
1070  REM ************************************************************************
1080 SUMSTDX = 0
1090 SDY = 0
1100 M = 0
1110  FOR I = 1 TO N
1120 SUMSTDX = ((X(I) - MX) © 2) + SUMSTDX
1130 SDY = ((Y(I) - MY) © 2) + SDY
1140 RD(I) = (((X(I) - MX) © 2) + ((Y(I) - MY) © 2)) © .5
1150 TR(I) = ( INT ((RD(I) + .005) * 100)) / 100
1160 M = RD(I) + M
1170  NEXT I
1180 DX = (SUMSTDX / (N - 1)) © .5
1190 DX = ( INT ((DX + .005) * 100)) / 100
1200 DY = (SDY / (N - 1)) © .5
1210 DY = ( INT ((DY + .005) * 100)) / 100
1220 SDSUM = ((DX © 2) + (DY © 2))
1230 SDSUM = ( INT ((SDSUM + .005) * 100)) / 100
1240 MRD = M / N
1250 MRD = ( INT ((MRD + .005) * 100)) / 100
1260  REM ***************************************
1270  REM *** DETERMINE MEDIAN RADIAL DISTANCE ***
1280  REM ***************************************
1290  FOR I = N TO 2 STEP  - 1
1300  FOR G = 1 TO N - 1
1310  IF RD(G) < RD(G + 1) GOTO 1350
1320 B = RD(G)
1330 RD(G) = RD(G + 1)
1340 RD(G + 1) = B
1350  NEXT G
```

```
LIST1360-1800

1360  NEXT I
1370  IF N > 2 * ( INT (N / 2)) GOTO 1400
1380 MDRD = ((RD(N / 2)) + (RD((N / 2) + 1))) / 2
1390  GOTO 1410
1400 MDRD = (RD((N / 2) + .5))
1410 MDRD = ( INT ((MDRD + .005) * 100)) / 100
1420  REM ************************************************************************
1430  REM *** DETERMINE DISTANCE, ANGLE, AND QUADRANT OF CENTER OF SHOT GROUP ***
1440  REM ************************************************************************
1450 DISTX = MX - XCM
1460 DTY = MY - YCM
1470  IF DISTX = 0 AND DTY = 0 GOTO 1600
1480 C = ((DISTX © 2) + (DTY © 2)) © .5
1490 C = ( INT ((C + .005) * 100)) / 100
1500  IF DISTX <  > 0 GOTO 1520
1510 T = 0:DG = 0: GOTO 1560
1520 T =  ATN ((DTY) / (DISTX))
1530 Q = 0
1540 DG = 57.29578 * T
1550 DG = ( INT ((DG + .005) * 100)) / 100
1560  IF (DISTX <  = 0) AND (DTY <  = 0) THEN Q = 3: IF Q = 3 THEN DG = 180 + DG
1570  IF (DISTX <  = 0) AND (DTY >  = 0) THEN Q = 2: IF Q = 2 THEN DG = 180 + DG
1580  IF (DISTX >  = 0) AND (DTY <  = 0) THEN Q = 4: IF Q = 4 THEN DG = 360 + DG
1590  IF (DISTX >  = 0) AND (DTY >  = 0) THEN Q = 1
1600  IF (DISTX = 0) OR (DTY = 0) THEN Q = 0
1610  IF ((DISTX > 0) AND (DTY = 0)) THEN DG = 0
1620  IF ((DISTX = 0) AND (DTY > 0)) THEN DG = 90
1630  IF ((DISTX < 0) AND (DTY = 0)) THEN DG = 180
1640  IF ((DISTX = 0) AND (DTY < 0)) THEN DG = 270
1650  REM ************************************************************************
1660  REM  *** PRINT THE RESULTS ON THE MONITOR OR PRINTER ***
1670  REM  *** NOTE:  TO ADAPT THIS PROGRAM TO OTHER MACHINES, PR#0 AND ***
1680  REM  *** PR#1 MUST BE CHANGED TO THEIR APPROPRIATE COUNTERPARTS ***
1690  REM ************************************************************************
1700  IF P$ <  > "Y" GOTO 1720
1710  PR# 1
1720  IF P$ <  > "Y" THEN  GOTO 1740
1730  GOTO 1750
1740  PRINT : PRINT : PRINT : PRINT : PRINT
1750  PRINT "SUBJECT CODE:  ";SC$: PRINT "NUMBER OF ROUNDS:  ";N
1760  PRINT "TARGET CENTER OF MASS:  ";"(";XCM;",";YCM;")"
1770  PRINT : PRINT
1780  PRINT "SHOT DATA:"
1790  PRINT
1800  IF P$ <  > "Y" THEN  GOTO 1830
```

```
LIST1810-

1810  PRINT "  SHOT";"   X   ";"   Y   ";"   RADIAL ERROR   ";"   RADIAL DISTANCE"
1820  GOTO 1850
1830  PRINT "SHOT";" X ";" Y ";" RAD ERR ";" RAD DIST "
1840  GOTO 1860
1850  PRINT  TAB( 23);"(TARGET CENTER)";"   (SHOT GROUP CENTER)"
1860  PRINT
1870  FOR I = 1 TO N
1880  IF P$ <  > "Y" THEN  GOTO 1910
1890  PRINT  TAB( 4);I; TAB( 6);X(I); TAB( 7);Y(I); TAB( 11);ROE(I); TAB( 17);TR(I)
1900  GOTO 1920
1910  PRINT  TAB( 3);I; TAB( 6);X(I); TAB( 9);Y(I); TAB( 14);ROE(I); TAB( 24);TR(I)
1920  NEXT I
1930  PRINT
1940  PRINT
1950  PRINT "MEAN X =  ";MX;"     MEAN Y =    ";MY
1960  PRINT "STANDARD DEVIATION X = ";DX;"   STANDARD DEVIATION Y =  ";DY
1970  PRINT "SUM OF X AND Y VARIANCES:  ";SDSUM
1980  IF P$ <  > "Y" GOTO 2000
1990  GOTO 2020
2000  PRINT "TO CONTINUE:  HIT ANY KEY"
2010  GET Z$
2020  PRINT : PRINT
2030  PRINT "SHOT GROUP DATA:": PRINT
2040  PRINT "RADIAL ERROR (TARGET CENTER):          MEAN =   ";MNRE;"    MEDIAN =    ";MEDR
2050  PRINT
2060  PRINT "RADIAL DISTANCE (SHOT GROUP CENTER):   MEAN =   ";MRD;"    MEDIAN =    ";MDRD
2070  PRINT : PRINT
2080  PRINT "COORDINATES OF CENTER OF SHOT GROUP :   (";MX;" , ";MY;")"
2090  PRINT "EXTREME SPREAD (SHOT ";LO;" AND SHOT ";HI;") =    ";EXSP
2100  PRINT "DISTANCE OF CENTER OF SHOT GROUP FROM CENTER OF MASS:   ";C
2110  PRINT "ANGLE OF SHOT GROUP FROM CENTER OF MASS IN DEGREES:   ";DG
2120  IF Q <  > 0 THEN  GOTO 2150
2130  PRINT "THE QUADRANT CANNOT BE DETERMINED"
2140  GOTO 2170
2150  PRINT "SHOT GROUP CENTER IS LOCATED IN QUADRANT:  ";Q
2160  IF P$ <  > "Y" THEN  GOTO 2200
2170  PRINT : PRINT : PRINT
2180  PRINT "*************************************************************************"
2190  PRINT : PRINT
2200  PRINT
2210  PR# 0
2220  NEXT D
2230  END
```